

# Excel VBA 常用物件介紹

## 9

CHAPTER

### 學習目標

- **Workbook** 物件常用的屬性、方法、事件
- **Worksheet** 物件常用的屬性、方法、事件
- **Sheet** 物件常用的屬性、方法、事件
- **Windows** 物件常用的屬性、方法、事件

## 9.1 Workbook 物件簡介與常用屬性

### 9.1.1 Workbook 物件簡介

在 Excel 應用程式中每一個開啟的 Workbook 活頁簿檔案，就是一個 Workbook 物件，多個 Workbook 物件就組合成一個 Workbooks 活頁簿集合(Collection)。目前作用中的活頁簿，可以用 `ActiveWorkbook` 來表示。另外，要指定目前 VBA 程式碼所在的活頁簿，則可以用 `ThisWorkbook` 來表示。如果要指定 Workbooks 活頁簿集合中特定的活頁簿，其語法如下：

語法：

`Workbooks(索引值)`  
或  
`Workbooks("活頁簿檔名")`

在程式中如果要宣告一個活頁簿物件，其語法如下：

語法：

```
Dim 活頁簿物件名稱 As Workbook
```

例如：宣告一個活頁簿物件 actWB，然後指定為目前作用中活頁簿，寫法如下：

```
Dim actWB As Workbook  
Set actWB = Application.ActiveWorkbook  
Set actWB = ActiveWorkbook '此為上一行的精簡寫法
```

## 9.1.2 Workbook 物件常用的屬性

### 1. Count 屬性

使用 Count 屬性可以取得活頁簿集中活頁簿的數量。例如：顯示活頁簿的總數量，寫法為：

```
MsgBox("活頁簿的數量：" & Workbooks.Count)
```

### 2. Name 屬性

使用 Name 屬性可以取得活頁簿的名稱。例如：顯示目前作用活頁簿的名稱：

```
MsgBox("目前活頁簿的名稱：" & ActiveWorkbook.Name)
```

### 3. FullName 屬性

使用 FullName 屬性可以取得活頁簿的完整名稱，含路徑和檔案名稱。例如：顯示目前作用活頁簿的完整名稱，寫法為：

```
MsgBox("目前活頁簿的完整名稱：" & ActiveWorkbook.FullName)
```

### 4. Path 屬性

使用 Path 屬性可取得活頁簿存放的路徑。例如：顯示目前作用活頁簿的路徑：

```
MsgBox("目前活頁簿的路徑：" & ActiveWorkbook.Path)
```

## 5. ActiveSheet 屬性

使用 ActiveSheet 屬性可以取得活頁簿中目前作用中的工作表。例如：顯示目前作用中工作表的名稱，寫法為：

```
MsgBox "目前工作表的名稱：" & ActiveSheet.Name
```

## 6. Saved 屬性

使用 Saved 屬性可以得知活頁簿是否已經儲存，活頁簿開檔後如果沒有修改內容 Saved 屬性值會為 True；如果內容有修改則 Saved 屬性值會被設定為 False。如果雖然有修改但不想儲存時，可以自行將 Saved 屬性值設為 True。例如：目前作用活頁簿內容有修改時就顯示訊息，寫法為：

```
If ActiveWorkbook.Saved = False Then
    MsgBox "這個活頁簿有修改過需要儲存！"
End If
```

## 7. Password 屬性

使用 Password 屬性可以設定和取得活頁簿的密碼，密碼為不超過 15 個字元的字串(有分大小寫)。例如：設定目前作用活頁簿的密碼，寫法為：



```
ActiveWorkbook.Password = InputBox("請設定活頁簿的密碼：")
```



### 實作 FileName : Workbook 屬性.xlsm

程式執行時會出現對話方塊，可以輸入目前活頁簿的密碼。輸入後會顯示目前活頁簿的數量、密碼、名稱、完整名稱和路徑的資訊。最後清除密碼，以避免忘記密碼時無法開啟檔案。


### ► 輸出要求



## ► 解題技巧

### Step 1 建立輸出入介面

1. 新增活頁簿並以「Workbook 屬性」為新活頁簿名稱。
2. 在工作表 1 中建立一個 ActiveX 命令按鈕



### Step 2 問題分析

1. 使用 InputBox 接受輸入的密碼，並指定給 Password 屬性。顯示 Password 屬性值時，因為密碼加密過已經非原輸入值，而且以\*字元顯示。為了避免忘記密碼無法開啟檔案，最後設 Password 屬性值為空字串來清除密碼。
2. 使用 Count、Name、FullName、Path 等屬性，來顯示活頁簿的相關資訊。

### Step 3 編寫程式碼

#### FileName: Workbook 屬性.xlsm (工作表 1 程式碼)



##### 01 Private Sub CommandButton1\_Click()

```

02   ActiveWorkbook.Password = InputBox("請設定活頁簿的密碼：")
03   MsgBox ("活頁簿的數量：" & Workbooks.Count)
04   MsgBox ("活頁簿的密碼：" & ActiveWorkbook.Password)
05   MsgBox ("目前活頁簿的名稱：" & ActiveWorkbook.Name)
06   MsgBox ("目前活頁簿的完整名稱：" & ActiveWorkbook.FullName)
07   MsgBox ("目前活頁簿的路徑：" & ActiveWorkbook.Path)
08   ActiveWorkbook.Password = "" '刪除密碼
09 End Sub
    
```

## ► 隨堂練習

程式執行時會顯示目前活頁簿的數量、名稱、路徑和是否需要儲存等資訊。若需要儲存時會出現對話方塊，詢問是否改為不要儲存，若按 <是> 鈕就設為已經儲存。



## 9.6 Worksheet 物件常用事件

### 9.6.1 建立 Worksheet 物件事件的步驟

要建立 Worksheet 物件事件的操作步驟如下所示：


1. 開啟工作表的程式碼視窗

在專案視窗要建立事件的工作表上快按兩下，就會開啟該工作表的程式碼視窗。



2. 建立工作表事件

在物件清單中選取 Worksheet 物件，此時會自動建立 Worksheet 物件的預設事件 SelectChange。如果要再繼續建立事件，可以由事件清單中選取要建立的事件。



### 9.6.2 Worksheet 物件常用的事件

1. Activate 事件

當選取工作表使成作用工作表時，就會觸動該工作表的 Activate 事件。例如：  
在 Activate 事件中，顯示 "選取 xxx 工作表" 訊息，其中 xxx 表示工作表名稱：

```
Private Sub Worksheet_Activate()  
    MsgBox "選取" & ActiveSheet.Name & "工作表", vbOKOnly, "Activate 事件"  
End Sub
```



## 便利貼

在 Workbook 物件的 SheetActivate 事件和 Worksheet 的 Activate 事件一樣，當工作表成作用工作表時都會被觸動。如果該處理對每個工作表都相同時，程式就寫在 SheetActivate 事件中。如果只針對某個工作表處理時，程式就寫在該工作表的 Activate 事件中。

## 2. Deactivate 事件

當選取其他工作表使原工作表成為非作用工作表時，就會觸動該工作表的 Deactivate 事件。例如：在 Deactivate 事件中，顯示 "離開工作表" 訊息的寫法：

```
Private Sub Worksheet_Deactivate ()  
    MsgBox "離開工作表", vbOKOnly, "Deactivate 事件"  
End Sub
```

## 3. Change 事件

當工作表變更儲存格內容時，就會觸動該工作表的 Change 事件。但是如果是因為重算而變更儲存格內容時，則不會觸動 Change 事件，若需要處理時可以寫在該工作表的 Calculate 事件。事件中 Target 參數代表變動的儲存格。例如：在 Change 事件中，顯示 "xxx 儲存格被改變"，其中 xxx 即改變內容的儲存格位置訊息，寫法如下：

```
Private Sub Worksheet_Change(ByVal Target as Range)  
    MsgBox Target.Address & "儲存格被改變", vbOKOnly, "Change 事件"  
End Sub
```

## 4. SelectionChange 事件

當工作表變更儲存格選取範圍時，就會觸動該工作表的 SelectionChange 事件。例如：在 SelectionChange 事件中，檢查是否是選擇第一欄內的一個儲存格？若是，則顯示 "選取了第一欄的一個儲存格" 訊息，寫法如下：

```
Private Sub Worksheet_SelectionChange(ByVal Target as Range)  
    If Target.Column = 1 And Target.Count = 1 Then  
        MsgBox "選取了第一欄的一個儲存格", vbOKOnly, "SelectionChange 事件"  
    End If  
End Sub
```

## 5. BeforeDoubleClick 事件

當在儲存格快按兩下時，就會觸動該工作表的 BeforeDoubleClick 事件。因為快按兩下會進入編輯狀態，可以將 Cancel 參數設為 True 避免進入編輯狀態。

```
Private Sub Worksheet_BeforeDoubleClick(ByVal Target as Range, Cancel As Boolean)
    Cancel = True
    MsgBox Target.Address & "儲存格被快按兩下"
End Sub
```



### 實作 FileName : Worksheet 事件.xlsm

使用者點選 A 欄內的一個儲存格時，會顯示對話方塊詢問是否設為粗體字？若按  鈕該儲存格設為粗體字。使用者修改時薪和上班時數欄內的資料時，會檢查資料是否正確，如果是下列情況會顯示提示訊息：輸入非數值、輸入低於 120 元的時薪、輸入高於 48 小時的上班時數。

#### ► 輸出要求

A	B
1 姓名	
2 素煩真	100
3 葉小差	140
4 一夜輸	130

A	B	C	D
1 姓名	時薪	上班時數	薪資
2 素煩真	120	42	5040
3 葉小差	140	58	8120
4 一夜輸	130	36	4680

#### ► 解題技巧


##### Step 1 建立輸出入介面

1. 新增活頁簿並以「Worksheet 事件」為新活頁簿名稱。

2. 工作表中建立如右的表格資料，D 欄輸入薪資公式。

	A	B	C	D
1	姓名	時薪	上班時數	薪資
2	素煩真	120	40	4800
3	葉小差	140	48	6720
4	一夜輪	130	36	4680

### Step 2 問題分析

1. 因為都是屬於工作表的事件，所以必須在工作表中建立事件程序。
2. 當選取工作表內的儲存格時會觸動 SelectionChange 事件，在事件程序中檢查選取的儲存格是否在 A 欄(Target.Column = 1)，以及是否只選取一個儲存格(Target.Count = 1)。如果符合就顯示 MsgBox 對話方塊，若按  鈕就設為粗體字(Target.Font.Bold = True)；否則就設為非粗體字。
3. 當修改工作表內的儲存格內容時會觸動 Change 事件，在事件程序中先檢查修改的儲存格是否在 B、C 欄，以及是否在第 2 列以上(Target.Row >= 2)。接著檢查輸入值是否為數值(IsNumeric(Target))，若是 B 欄資料輸入值必須大於 120，C 欄資料輸入值必須小於 48，否則就顯示提示訊息並設為預設值。

因為會用程式將儲存格設為預設值，此時會再觸動 SelectionChange 事件。為提高效率可以先將 Application 的 EnableEvents 屬性值設為 False，來暫停觸動事件，最後必須再設為 True，不然事件都不會觸動。

### Step 3 編寫程式碼

FileName: Worksheet 事件.xlsm (工作表 1 程式碼)

```
01 Private Sub Worksheet_SelectionChange(ByVal Target As Range)
```

```
02   If Target.Column = 1 And Target.Count = 1 Then
```

```
03     YorN = MsgBox("是否設為粗體字?", vbYesNo)
```

```
04     If YorN = vbYes Then
```

```
05       Target.Font.Bold = True
```

```
06     Else
```

```
07       Target.Font.Bold = False
```

```
08     End If
```

```
09   End If
```

```
10 End Sub
```

```
11
```



```

12 Private Sub Worksheet_Change(ByVal Target As Range)
13   Application.EnableEvents = False      '暫停觸動事件
14   If Target.Column = 2 And Target.Row >= 2 Then
15     If IsNumeric(Target) Then  '如果是數值
16       If Target.Value < 120 Then  '若輸入值小於 120
17         MsgBox "時薪不能低於 120 元!", vbOKOnly
18         Target.Value = 120      '設為預設值
19       End If
20     Else
21       MsgBox "請輸入數值!", vbOKOnly
22       Target.Value = 120      '設為預設值
23     End If
24   ElseIf Target.Column = 3 And Target.Row >= 2 Then
25     If IsNumeric(Target) Then  '如果是數值
26       If Target.Value > 48 Then  '若輸入值大於 48
27         MsgBox "每周工時不能超過 48 小時!", vbOKOnly
28         Target.Value = 48      '設為預設值
29       End If
30     Else
31       MsgBox "請輸入數值!", vbOKOnly
32       Target.Value = 48      '設為預設值
33     End If
34   End If
35   Application.EnableEvents = True      '開啟觸動事件
36 End Sub

```

► 隨堂練習


修改上面實例點選 A 欄內的一個儲存格時，有資料才會顯示對話方塊詢問是否設為粗體字？A 欄內輸入姓名後，B~D 欄會自動填入預設值和薪資計算公式；若是空白則清除 B~D 欄資料。

	A	B	C	D
1	姓名	時薪	上班時數	薪資
2	素煩真	120	40	4800
3	葉小差	140	48	6720
4	一夜輪	130	36	4680
5	史艷文	120	48	5760

## 9.7 Window 物件

### 9.7.1 Window 物件簡介

在 Excel 應用程式中可以同時開啟多個活頁簿，而每個活頁簿都有一個視窗(Window)，多個 Window 物件就組成一個 Windows 視窗集物件。在 VBA 中可以使用 Window 物件，來操作所屬的視窗。



目前作用中的視窗，可以用 ActiveWindow 來表示。如果要指定 Windows 視窗集合中特定的視窗，其語法如下：

語法：

Windows (索引值)  
或  
Windows ("活頁簿檔案名稱")

### 9.7.2 Window 物件常用的屬性

#### 1. WindowNumber 屬性

使用 WindowNumber 屬性可以取得視窗的編號，其值和視窗集合的索引值相同。例如：顯示目前作用視窗的編號，寫法為：

```
MsgBox "目前視窗的編號：" & ActiveWindow.WindowNumber
```

#### 2. Caption 屬性

Caption 屬性可以設定和取得視窗的名稱。例如：顯示目前作用視窗的名稱：

```
MsgBox "目前視窗的名稱：" & ActiveWindow.Caption
```

#### 3. WindowState 屬性

使用 WindowState 屬性可以設定視窗的顯示狀態，屬性值有：xlMaximized(視窗最大化)、xlMinimized(視窗最小化)、xlNormal (手動)三種。例如：將目前作用

視窗縮小到工作列，寫法為：

```
ActiveWindow.WindowState = xlMinimized
```

#### 4. EnableResize 屬性

使用 EnableResize 屬性可以設定視窗是否允許調整大小，屬性值有 True(可)、False(不可)。例如：設定目前作用視窗不能調整大小，寫法為：

```
ActiveWindow.EnableResize = False
```

#### 5. Zoom 屬性

使用 Zoom 屬性可以設定或取得視窗的顯示比例，屬性值的資料型態為 Variant，值如果為 50 表縮小一半、100 表正常大小、200 表放大一倍。例如：新增一個活頁簿時設定視窗顯示比例為 100%，寫法為：

```
Private Sub Workbook_Open()  
    ActiveWindow.Zoom = 100  
End Sub
```

#### 6. Height、Width 屬性

使用 Height、Width 屬性可以分別設定或取得視窗的高度和寬度，但 WindowState 屬性值必須是 xlNormal (手動)才能設定視窗大小。

#### 7. DisplayGridlines 屬性

使用 DisplayGridlines 屬性可以設定視窗內的工作表是否顯示格線，屬性值有 True(顯示)、False(不顯示)。例如：改變目前作用視窗格線的顯示狀態，寫法：

```
ActiveWindow.DisplayGridlines = Not(ActiveWindow.DisplayGridlines)
```

#### 8. VisibleRange 屬性

使用 VisibleRange 屬性可以取得指定視窗目前可視的儲存格範圍。例如：顯示目前作用視窗可視的儲存格範圍的位址，寫法為：

```
MsgBox ActiveWindow.VisibleRange.Address
```

設定目前工作表可以捲動的範圍為視窗目前可視的儲存格範圍，寫法為：

```
ActiveSheet.ScrollArea = ActiveWindow.VisibleRange.Address
```

## 9. SelectedSheets 屬性

使用 SelectedSheets 屬性可以取得指定視窗中所有選取的工作表(sheets 集合)。

例如：顯示目前作用視窗所有選取工作表的名稱，寫法為：

```
For Each sh In ActiveWindow.SelectedSheets
    MsgBox "選取" + sh.Name + "工作表"
Next
```

### 9.7.3 Window 物件常用的方法

#### 1. NewWindow 方法

使用 NewWindow 方法可以將指定視窗建立複本成為新的視窗。例如：將第一個視窗複製成新視窗，寫法為：

```
ActiveWorkbook.Windows(1).NewWindow
```

#### 2. Activate 方法

使用 Activate 方法可以將指定視窗成為作用視窗，該視窗會移到視窗集合的第一個。例如：將第三個視窗成為作用視窗，寫法為：

```
ActiveWorkbook.Windows(3).Activate '目前活頁簿的第三個視窗成為作用視窗
Application.Windows(3).Activate 'Excel 應用程式的第三個視窗成為作用視窗
```

#### 3. Close 方法

使用 Close 方法可以關閉指定的視窗，例如：關閉目前作用視窗寫法為：

```
ActiveWindow.Close
```



**實作** FileName : Total.xlsm

按  鈕會將活頁簿中所有工作表的各月出貨清單，全部合併到「合計」工作表中。

► 輸出要求

	A	B	C	D	E	F	G	H	I
1	日期	客戶	產品	數量	售價	金額	負責人		
2	2016/1/2	東方企業社	震動電鑽	2	2370	4740	樂樂	合併	
3	2016/1/8	齊美公司	抽排風機	1	5680	5680	芳美		
4	2016/1/14	東方企業社	直流電焊機	1	9800	9800	樂樂		
5	2016/1/21	峰源工業	乾濕吸塵器	3	5900	17700	芳美		
6	2016/1/26	台楠工業社	高壓沖洗機	2	3680	7360	芳美		
7	2016/1/28	雅洲工業社	平面砂輪機	5	1890	9450	樂樂		
8	2016/1/30	振達五金公司	熱風槍	2	2380	4760	芳美		
9									

	A	B	C	D	E	F	G	H
1	日期	客戶	產品	數量	售價	金額	負責人	
2	2016/1/2	東方企業社	震動電鑽	2	2370	4740	樂樂	
3	2016/1/8	齊美公司	抽排風機	1	5680	5680	芳美	
4	2016/1/14	東方企業社	直流電焊機	1	9800	9800	樂樂	
5	2016/1/21	峰源工業	乾濕吸塵器	3	5900	17700	芳美	
6	2016/1/26	台楠工業社	高壓沖洗機	2	3680	7360	芳美	
7	2016/1/28	雅洲工業社	平面砂輪機	5	1890	9450	樂樂	
8	2016/1/30	振達五金公司	熱風槍	2	2380	4760	芳美	
9	2016/2/4	齊美公司	沉水馬達	2	2499	4998	芳美	
10	2016/2/9	東方企業社	緊急照明燈	6	485	2910	樂樂	
11	2016/2/12	峰源工業	手提剪草機	3	2745	8235	芳美	
12	2016/2/16	台楠工業社	鏈鋸機	2	3680	4600	芳美	
13	2016/2/23	東方企業社	斜口鉗	12	199	2388	樂樂	
14	2016/3/1	東方企業社	籬笆剪	1	5988	5988	樂樂	
15	2016/3/6	振達五金公司	電動起子	2	4480	8960	芳美	
16	2016/3/10	全友五金社	打蠟機	6	3990	23940	樂樂	
17	2016/3/17	峰源工業	手提剪草機	3	2745	8235	芳美	
18	2016/3/26	台楠工業社	空壓機	2	7399	4600	芳美	
19	2016/3/26	台楠工業社	刻磨機	3	680	2040	芳美	
	合計							

► 解題技巧

Step 1 建立輸出入介面

1. 新增活頁簿並以「Total」為新活頁簿名稱。
2. 將範例 ch09 資料夾裡 Total 資料.xlsx 檔案中的工作表複製到活頁簿中，並在「1月」工作表中建立一個 ActiveX 命令按鈕控制項。

Step 2 問題分析

1. 可以使用 Add 方法新增一個工作表，但是要先確定「合計」工作表是否已經存在，如果已經存在就用 Clear 方法清除內容。

2. 想確認「合計」工作表是否已經存在時，可以先宣告一個 ws 工作表物件，再將「合計」工作表指定給物件。如果 ws 物件值為 Nothing，就表示「合計」工作表不存在。工作表不存在卻要指定時會產生錯誤，所以必須使用 On Error Resume Next 敘述來避免發生錯誤。
3. 因為是在「1月」工作表前新增工作表，所以「合計」工作表的索引值為 1，「1月」工作表索引值為 2 依此類推。
4. 複製資料時可以使用 Copy 方法，先複製「1月」工作表的標題列。然後用 For 迴圈，由第 2 個起到最後一個(索引值為 Worksheets.Count)工作表逐一複製資料到「合計」工作表。
5. 因為每個月份的資料長度不一，所以複製資料時可以使用 UsedRange.Rows.Count，來取得工作表有資料的列數。各月出貨清單用 Rows 屬性由第二列起取到最後一列，然後複製到「合計」工作表有資料的下一列。

### Step 3 編寫程式碼

#### FileName: Total.xlsm (工作表 1 程式碼)

##### 01 Private Sub CommandButton1\_Click()

```

02 Dim ws As Worksheet '宣告 ws 為工作表物件
03 On Error Resume Next '如果發生錯誤就執行下一行
04 Set ws = Worksheets("合計") '設 ws 為 合計 工作表
05 If ws Is Nothing Then '如果 ws 值為 Nothing
06     Worksheets(1).Select
07     Worksheets.Add '在第一個工作前面新增一個工作表
08     Worksheets(1).Name = "合計" '命名為 合計
09 Else
10     Worksheets("合計").Rows.Clear '清除 合計 工作表的內容
11 End If
12 On Error GoTo 0 '恢復錯誤處理
13 Worksheets(2).Activate
14 Rows(1).Copy Destination:=Worksheets(1).Range("A1") '複製標題到 合計 工作表
15 Dim r As Integer, r1 As Integer
16 For i = 2 To Worksheets.Count '從第 2 個工作表起
17     With Worksheets(i)
18         r = .UsedRange.Rows.Count '取得各月工作表的列數
19         r1 = Worksheets(1).UsedRange.Rows.Count '取得 合計 工作表的列數
20         '複製標題列以外的資料到 合計 工作表的空白列中
21         .Rows("2:" & r).Copy Destination:=Worksheets(1).Range("A" & r1 + 1)

```

22 End With

23 Next

24 End Sub

► 隨堂練習

修改上面實作使可以指定起始和終止月份，然後合併指定月份的資料到「合計」工作表。當起始月份大於終止月份，或輸入月份超出現有工作表數量時，會讓使用者重新輸入月份。



合併 2~3 月份  
資料 →

	A	B	C	D	E	F	G
1	日期	客戶	產品	數量	售價	金額	負責人
2	2016/2/4	齊美公司	沉水馬達	2	2499	4998	芳美
3	2016/2/9	東方企業社	緊急照明燈	6	485	2910	樂樂
4	2016/2/12	峰源工業	手提剪草機	3	2745	8235	芳美
5	2016/2/16	台楠工業社	鏈鋸機	2	3680	4600	芳美
6	2016/2/23	東方企業社	斜口鉗	12	199	2388	樂樂
7	2016/3/1	東方企業社	籬笆剪	1	5988	5988	樂樂
8	2016/3/6	振達五金公司	電動起子	2	4480	8960	芳美
9	2016/3/10	全友五金社	打蠟機	6	3990	23940	樂樂
10	2016/3/17	峰源工業	手提剪草機	3	2745	8235	芳美
11	2016/3/26	台楠工業社	空壓機	2	7399	4600	芳美
12	2016/3/26	台楠工業社	刻磨機	3	680	2040	芳美
13							